NAG Library Routine Document

F07HNF (ZPBSV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07HNF (ZPBSV) computes the solution to a complex system of linear equations

$$AX = B,$$

where A is an n by n Hermitian positive definite band matrix of bandwidth $(2k_d + 1)$ and X and B are n by r matrices.

2 Specification

```
SUBROUTINE F07HNF (UPLO, N, KD, NRHS, AB, LDAB, B, LDB, INFO)

INTEGER                 N, KD, NRHS, LDAB, LDB, INFO
COMPLEX (KIND=nag_wp) AB(LDAB,*), B(LDB,*)
CHARACTER(1)            UPLO
```

The routine may be called by its LAPACK name ***zpbsv***.

3 Description

F07HNF (ZPBSV) uses the Cholesky decomposition to factor A as $A = U^{\mathrm{H}}U$ if UPLO = 'U' or $A = LL^{\mathrm{H}}$ if UPLO = 'L', where U is an upper triangular band matrix, and L is a lower triangular band matrix, with the same number of superdiagonals or subdiagonals as A. The factored form of A is then used to solve the system of equations $AX = B$.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER(1) *Input*

On entry: if UPLO = 'U', the upper triangle of A is stored.

If UPLO = 'L', the lower triangle of A is stored.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the number of linear equations, i.e., the order of the matrix A.

Constraint: N $\geq$ 0.

3: KD – INTEGER *Input*

On entry: k_d, the number of superdiagonals of the matrix A if UPLO = 'U', or the number of subdiagonals if UPLO = 'L'.

Constraint: KD ≥ 0.

4: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrix B.

Constraint: NRHS ≥ 0.

5: AB(LDAB,$*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1, \text{N})$.

On entry: the upper or lower triangle of the Hermitian band matrix A.

The matrix is stored in rows 1 to $k_d + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in AB$(k_d + 1 + i - j, j)$ for $\max(1, j - k_d) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in AB$(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_d)$.

On exit: if INFO = 0, the triangular factor U or L from the Cholesky factorization $A = U^{\text{H}}U$ or $A = LL^{\text{H}}$ of the band matrix A, in the same storage format as A.

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07HNF (ZPBSV) is called.

Constraint: LDAB $\geq$ KD + 1.

7: B(LDB,$*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.

On entry: the n by r right-hand side matrix B.

On exit: if INFO = 0, the n by r solution matrix X.

8: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07HNF (ZPBSV) is called.

Constraint: LDB $\geq \max(1, \text{N})$.

9: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, the leading minor of order i of A is not positive definite, so the factorization could not be completed, and the solution has not been computed.

7 Accuracy

The computed solution for a single right-hand side, $\hat{x}$, satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the ***machine precision***. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A)\frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \left\|A^{-1}\right\|_1 \|A\|_1$, the condition number of A with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

F07HPF (ZPBSVX) is a comprehensive LAPACK driver that returns forward and backward error bounds and an estimate of the condition number. Alternatively, F04CFF solves $Ax = b$ and returns a forward error bound and condition estimate. F04CFF calls F07HNF (ZPBSV) to solve the equations.

8 Further Comments

When $n \gg k$, the total number of floating point operations is approximately $4n(k+1)^2 + 16nkr$, where k is the number of superdiagonals and r is the number of right-hand sides.

The real analogue of this routine is F07HAF (DPBSV).

9 Example

This example solves the equations

$$Ax = b,$$

where A is the Hermitian positive definite band matrix

$$A = \begin{pmatrix} 9.39 & 1.08 - 1.73i & 0 & 0 \\ 1.08 + 1.73i & 1.69 & -0.04 + 0.29i & 0 \\ 0 & -0.04 - 0.29i & 2.65 & -0.33 + 2.24i \\ 0 & 0 & -0.33 - 2.24i & 2.17 \end{pmatrix}$$

and

$$b = \begin{pmatrix} -12.42 + 68.42i \\ -9.93 + 0.88i \\ -27.30 - 0.01i \\ 5.31 + 23.63i \end{pmatrix}.$$

Details of the Cholesky factorization of A are also output.

9.1 Program Text

```
    Program f07hnfe

!     F07HNF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
```

```
      Use nag_library, Only: nag_wp, x04dff, zpbsv
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
      Character (1), Parameter         :: uplo = 'U'
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, j, kd, ldab, n
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,:), b(:)
      Character (1)                    :: clabs(1), rlabs(1)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F07HNF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd
      ldab = kd + 1
      Allocate (ab(ldab,n),b(n))

!     Read the upper or lower triangular part of the band matrix A
!     from data file

      If (uplo=='U') Then
        Read (nin,*)((ab(kd+1+i-j,j),j=i,min(n,i+kd)),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)((ab(1+i-j,j),j=max(1,i-kd),i),i=1,n)
      End If

!     Read b from data file

      Read (nin,*) b(1:n)

!     Solve the equations Ax = b for x
!     The NAG name equivalent of zpbsv is f07hnf
      Call zpbsv(uplo,n,kd,1,ab,ldab,b,n,info)

      If (info==0) Then

!       Print solution

        Write (nout,*) 'Solution'
        Write (nout,99999) b(1:n)

!       Print details of factorization

        Write (nout,*)
        Flush (nout)

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        If (uplo=='U') Then
          Call x04dff(n,n,0,kd,ab,ldab,'Bracketed','F7.4','Cholesky factor U', &
            'Integer',rlabs,'Integer',clabs,80,0,ifail)
        Else If (uplo=='L') Then
          Call x04dff(n,n,kd,0,ab,ldab,'Bracketed','F7.4','Cholesky factor L', &
            'Integer',rlabs,'Integer',clabs,80,0,ifail)
        End If

      Else
        Write (nout,99998) 'The leading minor of order ', info, &
          ' is not positive definite'
      End If

99999 Format ((3X,4(' (',F7.4,',',F7.4,')':)))
99998 Format (1X,A,I3,A)
    End Program f07hnfe
```

9.2 Program Data

```
F07HNF Example Program Data

   4             1                                          :Values of N and KD

 (  9.39, 0.00) (  1.08,-1.73)
               (  1.69, 0.00) ( -0.04, 0.29)
                              (  2.65, 0.00) ( -0.33, 2.24)
                                             (  2.17, 0.00) :End of matrix A

 (-12.42,68.42) ( -9.93, 0.88) (-27.30,-0.01) (  5.31,23.63) :End of vector b
```

9.3 Program Results

```
 F07HNF Example Program Results

 Solution
    (-1.0000, 8.0000) ( 2.0000,-3.0000) (-4.0000,-5.0000) ( 7.0000, 6.0000)

 Cholesky factor U
                    1                 2                 3                 4
 1  ( 3.0643, 0.0000) ( 0.3524,-0.5646)
 2                    ( 1.1167, 0.0000) (-0.0358, 0.2597)
 3                                      ( 1.6066, 0.0000) (-0.2054, 1.3942)
 4                                                        ( 0.4289, 0.0000)
```